# $Id$
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=xorg-xfd
pkgver=1.1.2
pkgrel=2
pkgdesc="Displays all the characters in a font using either the X11 core protocol or libXft2"
arch=(x86_64)
url="https://xorg.freedesktop.org/"
license=('custom')
depends=('libxaw' 'fontconfig' 'libxft' 'libxrender' 'libxmu')
makedepends=('xorg-util-macros')
source=(https://xorg.freedesktop.org/archive/individual/app/xfd-${pkgver}.tar.bz2)
sha512sums=('3a158f697dd40c5b650b3d80ad1692d668d9d9bcf586a437fd3dcebf9e8fffb773855df290260995464d5f36cb22e768d1d71de8e61d7cca80eaa4d25c029745')

build() {
  cd "${srcdir}/xfd-${pkgver}"
  ./configure --prefix=/usr
  make
}

package() {
  cd "${srcdir}/xfd-${pkgver}"
  make DESTDIR="${pkgdir}" install
  install -m755 -d "${pkgdir}/usr/share/licenses/${pkgname}"
  install -m644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/"
}
